The Lotka-Volterra predator prey model.

Let $x(t)$ be the population of the prey at time t
Let $y(t)$ be the population of the predators at time t

```
> restart;
  de1 := diff( x(t),t) = alpha1*x(t) - beta1*x(t)*y(t);
```

$$de1 := \frac{d}{dt} x(t) = \alpha 1\, x(t) - \beta 1\, x(t)\, y(t) \quad (1)$$

```
> de2 := diff( y(t),t) = beta2*x(t)*y(t) - alpha2*y(t);
```

$$de2 := \frac{d}{dt} y(t) = \beta 2\, x(t)\, y(t) - \alpha 2\, y(t) \quad (2)$$

```
> solve( {rhs(de1)=0,rhs(de2)=0}, {x(t),y(t)} );
```

$$\{x(t) = 0, y(t) = 0\}, \left\{x(t) = \frac{\alpha 2}{\beta 2}, y(t) = \frac{\alpha 1}{\beta 1}\right\} \quad (3)$$

```
> alpha1 := 0.1;
  beta1 := 0.1;
  beta2 := 0.02;
  alpha2 := 0.05;
```

$$\alpha 1 := 0.1$$
$$\beta 1 := 0.1$$
$$\beta 2 := 0.02$$
$$\alpha 2 := 0.05 \quad (4)$$

```
> F := dsolve( {de1,de2,x(0)=1,y(0)=0.2}, numeric );
```

$$F := \textbf{proc}(x\_rkf45) \ \ldots \ \textbf{end proc} \quad (5)$$

```
> F(0.0);
```

$$[t = 0., x(t) = 1., y(t) = 0.200000000000000] \quad (6)$$

```
> F(0.1);
```

$$[t = 0.1, x(t) = 1.00803510126440, y(t) = 0.199402498979488] \quad (7)$$

```
> F(2.0);
```

$$[t = 2.0, x(t) = 1.17484203973331, y(t) = 0.188993456628175] \quad (8)$$

```
> F(-2.0);
```

$$[t = -2.0, x(t) = 0.853224170436178, y(t) = 0.213009874075132] \quad (9)$$

```
> with(plots):
> odeplot( F, [t,x(t)], t=0..100 );
```




```
> odeplot( F, [[t,x(t)],[t,y(t)]], t=0..200, color=[blue,black], legend=
  ["prey","predators"] );
```




```
> ?odeplot
```

Animation, in 3D, etc.

```
> odeplot( F, [t,x(t),y(t)], t=0..200, thickness=2, axes=frame);
```




```
> with(DEtools):
> DEplot( {de1,de2}, [x(t),y(t)], t=0..100, [[x(0)=1,y(0)=0.2],[x(0)=5,y(0)=1]],
  linecolor=blue, numpoints=200, dirfield=100, arrows=comet );
```



What happens if we modify the differential equation for the predator to be

```
> de2 := diff(y(t),t) = -beta2*x(t)-alpha2*y(t);
```

$$de2 := \frac{d}{dt}\, y(t) = -0.02\, x(t) - 0.05\, y(t) \quad (10)$$

Go back and experiment with that

```
> restart;
  interface(imaginaryunit=_i);
```

$$I \quad (11)$$

```
> I^2;
```

$$I^2 \quad (12)$$

The SIR (Kermack McKendrick) virus spread model

S(t) is the number of succeptibles, I(t) is the number of infecteds, R(t) is the number of recovereds

```
> deS := diff(S(t),t) = -beta*S(t)*I(t);
  deI := diff(I(t),t) = beta*S(t)*I(t)-alpha*I(t);
  deR := diff(R(t),t) = alpha*I(t);
```

$$deS := \frac{d}{dt} S(t) = -\beta S(t) I(t)$$

$$deI := \frac{d}{dt} I(t) = \beta S(t) I(t) - \alpha I(t)$$

$$deR := \frac{d}{dt} R(t) = \alpha I(t) \quad \mathbf{(13)}$$

```
> beta := 0.3;
```

$$\beta := 0.3 \quad \mathbf{(14)}$$

```
> alpha := 0.1;
```

$$\alpha := 0.1 \quad \mathbf{(15)}$$

The initial values S(0)=0.99, I(0)=0.01, R(0)=0 mean 1% are infected, 99% are succeptible, and none have recovered at time 0.

```
> F := dsolve( {deS, deI, deR, S(0)=0.99, I(0)=0.01, R(0)=0}, {S(t),I(t),R(t)}, numeric );
```

$F :=$ **proc**(x_rkf45) ... **end proc** **(16)**

```
> with(plots):
> odeplot( F, [[t,S(t)],[t,I(t)],[t,R(t)]], t=0..100, numpoints=200 );
```



The plot shows that the virus is epidemic, that is, growing since I(t) (in blue) is increasing at time t=0.

We can't take the $\lim_{t \to \infty} S(t)$ to determine what happens to the survivors but the following shows that at t=300 the number of infecteds is nearly 0 and hence the virus has died out and the number of survivors is 5.88%.

```
> F(300);
```

$[t = 300., I(t) = 2.35201393323179\ 10^{-9}, R(t) = 0.941202616250399, S(t)$ **(17)**

$= 0.0587973813975870]$

Reexecute the above using

```
> beta := 0.1;
  alpha := 0.3;
```

$$\beta := 0.1$$

$$\alpha := 0.3 \quad \textbf{(18)}$$

```
> F := dsolve( {deS, deI, deR, S(0)=0.99, I(0)=0.01,R(0)=0}, {S(t),I(t),
  R(t)}, numeric );
```

$F :=$ **proc**(x_rkf45) ... **end proc** **(19)**

```
> odeplot( F, [[t,S(t)],[t,I(t)],[t,R(t)]], t=0..200, view=[0..200,0.
  .0.05], numpoints=200 );
```